# Cover Page for Statistical analysis plan

| | |
|---|---|
| Sponsor name: | Novo Nordisk A/S |
| NCT number | NCT02229851 |
| Sponsor trial ID: | NN8640-4054 |
| Official title of study: | A multicentre, multinational, randomised, parallel-group, placebo-controlled (double blind) and active-controlled (open) trial to compare the efficacy and safety of once weekly dosing of NNC0195-0092 with once weekly dosing of placebo and daily Norditropin® FlexPro® in adults with growth hormone deficiency for 35 weeks, with a 53-week extension period |
| Document date: | 27 November 2017 |

### 16.1.9 Documentation of statistical methods

# List of contents



*Redacted statistical analysis plan*
*Includes redaction of personal identifiable information only.*

.

**Statistical Analysis Plan**

# Trial ID: NN8640-4054

# REAL 1

**Author:**

[redacted]

# Table of contents

**Page**

# List of abbreviations

| | |
|---|---|
| AGHD | adult growth hormone deficiency |
| DM | diabetic mellitus |
| GH | growth hormone |
| GHD | growth hormone deficiency |
| hGH | human growth hormone |

# 1 Introduction

## 1.1 Trial information

NNC0195-0092 is a novel long-acting human growth hormone (hGH) derivative designed for once weekly administration in children with growth hormone deficiency (GHD) and adults with GHD (AGHD). NNC0195-0092 is a hGH derivative, consisting of a growth hormone (GH) back bone with a single point mutation to the amino acid backbone to which a non-covalent albumin binding moiety has been attached.

**Objective(s) and endpoint(s):**

**Primary objective**

- To demonstrate the efficacy of once weekly dosing of NNC0195-0092 compared to placebo after 34 weeks of treatment in adults with growth hormone deficiency

**Secondary objectives**

- To evaluate the clinical safety of once weekly dosing of NNC0195-0092 during 34 weeks of treatment in adults with growth hormone deficiency
- To evaluate the efficacy and safety of NNC0195-0092 for up to 86 weeks of treatment in adults with growth hormone deficiency (i.e. during the main and extension periods of the trial)

**Primary endpoint**

- Change from baseline to end of main treatment period (Week 34) in truncal fat percentage

**Key secondary endpoints for efficacy**

Changes from baseline to end of main treatment period (Week 34) in the following key variables will be used to address the primary objective

- Truncal fat mass (kg)
- Truncal lean body mass (kg)

**Key secondary endpoints for safety**

The following key endpoints will be used to support the secondary objectives of evaluation of safety in both the main (up to week 35) and extension (up to week 88) trial periods (including follow-up visits/washout periods):

- Incidence of adverse events, including injection site reactions
- Occurrence of anti-NNC0195-0092 antibodies

### Trial design

This is a multicentre, multinational, randomised, parallel-group, placebo-controlled (double blind) and active-controlled (open) trial. The trial will compare the efficacy and safety of once weekly dosing of NNC0195-0092 with once weekly dosing of placebo and daily dosing of Norditropin® FlexPro® in adults with GHD during a 35-week period (8 week dose titration, 26 week fixed dose treatment [34 weeks treatment period] followed by 1 week washout), followed by a 53-week open label extension period (8 week dose titration, 44 week fixed dose treatment [52 weeks treatment period] followed by 1 week washout). After the main trial period placebo subjects will be switched to NNC0195-0092 treatment and Norditropin® FlexPro® subjects will be randomised 1:1 to NNC0195-0092 or Norditropin® FlexPro®.




**Figure 1–1 Schematic trial overview**

## 1.2 Scope of the statistical analysis plan

This SAP version 2.0 is based on the protocol REAL 1, version 6.0. The SAP describes in details the updated planned analysis of the TRIM-AGHD based endpoints and the reason for the update of the analysis and a detailed description of the planned analysis of supportive secondary endpoint body weight.

# 2 Statistical considerations

From the protocol, section 17.4.1 Supportive secondary endpoints, page 91 , updated analysis text is written in italic:

Changes in scores of TRIM-AGHD (total and individual domain scores), SF-36 (physical and health component summary scores and individual domain scores) from baseline to the 8, 25, and 34 week's measurements will be analysed using a mixed model for repeated measurements (MMRM), with treatment, GHD onset type, sex, region, DM and sex by region by DM interaction as factors and baseline value as a covariate, all nested within week as a factor. An unstructured covariance matrix will be used to describe the variability for the repeated measurements for a subject. From the MMRM the treatment differences at Week 34 between NNC0195-0092 and placebo and between NNC0195-0092 and Norditropin® FlexPro® will be estimated and the corresponding 95% CIs and p-values will be calculated for each endpoint. Subjects without post-randomisation data for the analysed endpoint will not be included in the analysis. *As a sensitivity analysis, the analysis of TRIM-AGHD endpoints will be repeated restricted to subjects who answered the new questionnaire at their visits (i.e. subjects with the following introductory text in the questionnaire "The following questions are about how your growth hormone deficiency (GHD) impacts your functioning and well-being. Please tick the response box that most closely represents your CURRENT EXPERIENCE with GHD. Please tick only one response box for each question. Remember there are no right or wrong answers to these questions").*

From the protocol, section 17.4.1 Supportive secondary endpoints, page 91, added text is written in italic:

Changes in IGF-I SDS and IGFBP-3 SDS from baseline to the 1, 3, 5, 7, 9, 16, 25, 33 and 35 week's measurements will be analysed using a mixed model for repeated measurements (MMRM), with treatment, GHD onset type, sex, region, DM and sex by region by DM interaction as factors and baseline value as a covariate, all nested within week as a factor. An unstructured covariance matrix will be used to describe the variability for the repeated measurements for a subject. From the MMRM the treatment differences at Week 34 between NNC0195-0092 and placebo and between NNC0195-0092 and Norditropin® FlexPro® will be estimated and the corresponding 95% CIs and p-values will be calculated for each endpoint. Subjects without post-randomisation data for the analysed endpoint will not be included in the analysis.

*Changes in body weight from baseline to the 2, 8, 16, 25, 33 and 35 week's measurements will be analysed using a mixed model for repeated measurements (MMRM), with treatment, GHD onset type, sex, region, DM and sex by region by DM interaction as factors and baseline value as a covariate, all nested within week as a factor. An unstructured covariance matrix will be used to describe the variability for the repeated measurements for a subject. From the MMRM the treatment differences at Week 34 between NNC0195-0092 and placebo and between NNC0195-*

*0092 and Norditropin® FlexPro® will be estimated and the corresponding 95% CIs and p-values will be calculated for the endpoint. Subjects without post-randomisation data for the analysed endpoint will not be included in the analysis.*

# 3 Changes to the statistical analyses planned in the protocol

## 3.1 TRIM-AGHD

The introductory text of the TRIM-AGHD questionnaire has been updated after the finalisation of protocol version 2. For a few of the first enrolled subjects the updated version was not available at their randomisation visit (baseline visit) and they will answer the old questionnaire throughout the trial. The update of the introductory text consists of the deletion of the sentence "If you have other health conditions, please think only about your GHD when answering these questions." from the following "The following questions are about how your growth hormone deficiency (GHD) impacts your functioning and well-being. Please tick the response box that most closely represents your CURRENT EXPERIENCE with GHD. If you have other health conditions, please think only about your GHD when answering these questions. Please tick only one response box for each question. Remember there are no right or wrong answers to these questions."

In order to investigate whether the difference in the introductory text in the questionnaire has had any impact on the results of the analyses, all analyses of the TRIM-AGHD endpoints will be repeated restricted to subjects who answered the new questionnaire.

## 3.2 Body weight

A detailed description of planned analysis of supportive secondary endpoint body weight was not included in the protocol.

Sample size calculations protocol version 1

```
 proc power;
       twosamplemeans
       alpha=0.05
       dist=normal
       test=diff
       meandiff=2.5
       nulldiff=0
           gweights= (2 1)
       sides=2
       stddev=4.5
       power=0.8 to 0.95 by 0.05
       ntotal=.;
       ods output output=supres;
   run;


proc power;
   twosamplemeans
       alpha=0.05
       dist=normal
       ci=diff
       halfwidth=1.3 to 1.5 by 0.1
       probtype=unconditional
       sides=2
       stddev=4.5
       probwidth=.
       npergroup=104;
   ods output output=conf;
run;
data conf2;
   set conf ;
   label npergroup='N per trial arm (0092 or active comparator)'
       probwidth='Probability of halfwidth of 95% CI <= halfwidth parameter'
       halfwidth='Halfwidth parameter';
run;



title 'For comparison wrt chosen half width evaluation: changed Probtype to conditional';
proc power;
   twosamplemeans
       alpha=0.05
       dist=normal
       ci=diff
       halfwidth=1.3 to 1.5 by 0.1
       probtype=conditional
       sides=2
       stddev=4.5
       probwidth=.
       npergroup=104;
```

```
run;

title;

data supres2;
   set supres;
   n_treat=ntotal*2/3;
   ntotal_incl_act_comp=ntotal+n_treat;
   label nominalpower='Power'
      n_treat = 'N(0092 arm)'
      ntotal = 'Ntotal (2 arm trial)'
      ntotal_incl_act_comp='Ntotal (incl. active comp.)';
run;


ods rtf file="aghd2.rtf" style=minimal;

title 'Superiority test (0092 vs placebo) (diff=2.5%, SD=4.5%) ';
title2 'Active comparator (if included) only used for sec. analysis so no impact on power
evaluation';
proc print data=supres2 label noobs;
   var nominalpower n_treat  ntotal ntotal_incl_act_comp;


run;

title 'Sample size calculation for length of confidence interval (CI), comparison between 0092
and active comparato, SD=4.5%';
proc print data=conf2 label noobs;
   var probwidth halfwidth npergroup;
run;


ods rtf close;
quit;
```

The POWER Procedure
Two-Sample t Test for Mean Difference

Fixed Scenario Elements

| | |
|---|---|
| Distribution | Normal |
| Method | Exact |
| Number of Sides | 2 |
| Null Difference | 0 |
| Alpha | 0.05 |
| Mean Difference | 2.5 |
| Standard Deviation | 4.5 |
| Group 1 Weight | 2 |
| Group 2 Weight | 1 |

Computed N Total

| Index | Nominal Power | Actual Power | N Total |
|---|---|---|---|
| 1 | 0.80 | 0.802 | 117 |
| 2 | 0.85 | 0.856 | 135 |
| 3 | 0.90 | 0.902 | 156 |
| 4 | 0.95 | 0.951 | 192 |

The POWER Procedure
Confidence Interval for Mean Difference

Fixed Scenario Elements

| | |
|---|---|
| Distribution | Normal |
| Method | Exact |
| Number of Sides | 2 |
| Alpha | 0.05 |
| Standard Deviation | 4.5 |
| Sample Size Per Group | 104 |
| Prob Type | Unconditional |

Computed Prob(Width)

| Index | Half-Width | Prob (Width) |
|---|---|---|
| 1 | 1.3 | 0.879 |
| 2 | 1.4 | 0.997 |
| 3 | 1.5 | >.999 |

For comparison wrt chosen half width evaluation: changed Probtype to conditional 3

[redacted] [redacted]

The POWER Procedure
Confidence Interval for Mean Difference

Fixed Scenario Elements

| | |
|---|---|
| Distribution | Normal |
| Method | Exact |
| Number of Sides | 2 |
| Alpha | 0.05 |
| Standard Deviation | 4.5 |
| Sample Size Per Group | 104 |
| Prob Type | Conditional |

Computed Prob(Width)

| Index | Half-Width | Prob (Width) |
|---|---|---|
| 1 | 1.3 | 0.877 |
| 2 | 1.4 | 0.997 |
| 3 | 1.5 | >.999 |

Superiority test (0092 vs placebo) (diff=2.5%, SD=4.5%)
[redacted] 4
Active comparator (if included) only used for sec. analysis so no impact on power evaluation

| Power | N(0092 arm) | Ntotal (2 arm trial) | Ntotal (incl. active comp.) |
|---|---|---|---|
| 0.80 | 78 | 117 | 195 |
| 0.85 | 90 | 135 | 225 |
| 0.90 | 104 | 156 | 260 |
| 0.95 | 128 | 192 | 320 |

Sample size calculation for length of confidence interval (CI), comparison between 0092 and active comparato, SD=4.5% 5

[redacted]

[redacted]

| Probability of halfwidth of 95% CI <= halfwidth parameter | Halfwidth parameter | N per trial arm (0092 or active comparator) |
|---|---|---|
| 0.879 | 1.3 | 104 |
| 0.997 | 1.4 | 104 |

---

Sample size calculations amendment

program 4054_mi_sample_size_nobase_extra_v3.sas

* only 50% of the 7% dropout have second dxa;

```
%let nsim=10000; * Number of simulations;
%let sigma=4.5;  * Anticipated residual SD of percentage truncal fat;
%let mu=-2.5;      * Anticipated percentage truncal fat difference;
%let n1=112;       *Subjects peractive treatment  arm;
%let n2=56;       *Subjects per placebo arm;
%let sigma_base=10;
%let dropout=0.071;


data a;
 n=&n1.;

do nsim=1 to &nsim.;
do j=1 to &n2;


    do i=1 to 2;

 treat='Xhigh';
 y=&mu+&sigma.*rannor(3);

 if ranuni(5)<&dropout  then dropout=ranuni(5)*30;
 else dropout=100;
 output;

end;
treat='Xlow';
y=&sigma.*rannor(3);

 if ranuni(5)<&dropout  then dropout=ranuni(5)*30;
```

```
 else dropout=100;
output;
end;
end;
run;



proc sql;
    create table dropout as
        select nsim, treat, sum(dropout<100) as ndrop, n(y) as n, calculated ndrop/calculated n
as rate from a
        group by nsim, treat;
    select  treat, avg(rate) from dropout
        group by  treat;
    quit;




data b;
    set a;
    if dropout<100 then do;
        if ranbin(123, 1, 0.5)=0 then dropout=0;* some dropout subjects will not have second
dxa;
        dropval=y*dropout/34 ;
        y=.;
        end;
run;


ods listing exclude all;

proc MI data = b(where=(treat=:'Xlow' or dropout<100)) out = dataOutReg1 seed = 34247
nimpute = 100;
by nsim;
class treat;
var   treat  y;
monotone regression (y=  treat );
run;

ods listing;

data gentag;
    set  b(where=(treat =:'Xhigh' and dropout=100));
    by nsim;
    do _imputation_=1 to 100;

    output;
    end;
run;
```

```
proc print data=dataOutReg1 ;where nsim=10 and _imputation_=10;
run;


data datainReg2 ;
   set dataOutReg1 ;by nsim;
   if dropout<100 then y=y*(34-dropout)/34 + dropval;
run;

proc print data=datainReg2 ;where nsim=10 and _imputation_=10;
run;


data datainreg3;
   set datainReg2 gentag;
run;

proc sort data=datainreg3; by nsim _imputation_;
run;



proc sql;
   create table nodata as
       select nsim, treat, sum(dropout=0) as nodata, n(dropout) as n, calculated
nodata/calculated n as rate from datainreg3
      group by nsim, treat;
   select  treat, avg(rate) from  nodata
      group by  treat;
   quit;


ods listing exclude all;
proc mixed data = datainreg3;
by nsim _imputation_;
class treat ;
model y =    treat ;
lsmestimate treat 'treatdif' 1-1/CL;
ods output LSMestimates = estMI ;
run;


proc mianalyze data = estMI;
by nsim label;
modeleffects Estimate ;
stderr stderr;
ods output ParameterEstimates = MIana;
run;

ods listing;

proc sql;
```

```
    select sum(probt<0.05) as n_signif, n(probt) as n , avg(estimate) as mean_estimate from
miana;
    quit;
```

---

OUTPUT

dropout rate:



| treat | |
|---|---|
| Xhigh | 0.071368 |
| Xlow | 0.071057 |

proportion of subjects without post-randomization data



| treat | |
|---|---|
| Xhigh | 0.035579 |
| Xlow | 0.035846 |

simulated power(100*n_signif/n) and mean estimate



| n_signif | n | mean_ estimate |
|---|---|---|
| 8921 | 10000 | -2.35994 |

---

program 4054_mi_sample_size_nobase_extra_v4.sas

```
*15% dropout rate and all dropout are without post-randomisation data;

%let nsim=10000; * Number of simulations;
%let sigma=4.5;  * Anticipated residual SD of percentage truncal fat;
%let mu=-2.5;       * Anticipated percentage truncal fat difference;
%let n1=112;       *Subjects peractive treatment  arm;
```

```
%let n2=56;       *Subjects per placebo arm;
%let sigma_base=10;
%let dropout=0.151;


data a;
 n=&n1.;


do nsim=1 to &nsim.;
do j=1 to &n2;


   do i=1 to 2;

 treat='Xhigh';
 y=&mu+&sigma.*rannor(3);
 if ranuni(5)<&dropout  then dropout=ranuni(5)*30;
 else dropout=100;
 output;

end;
treat='Xlow';
y=&sigma.*rannor(3);
 if ranuni(5)<&dropout  then dropout=ranuni(5)*30;
 else dropout=100;
output;
end;
end;
run;



proc sql;
   create table dropout as
      select nsim, treat, sum(dropout<100) as ndrop, n(y) as n, calculated ndrop/calculated n
as rate from a
      group by nsim, treat;
   select  treat, avg(rate) from dropout
      group by  treat;
   quit;




data b;
   set a;
   if dropout<100 then do;
      dropout=0;*change from V3 program: dropout subjects will not have second dxa;
      dropval=y*dropout/34 ;
      y=.;
      end;
run;
```

```
ods listing exclude all;

proc MI data = b(where=(treat=:'Xlow' or dropout<100)) out = dataOutReg1 seed = 34247
nimpute = 100;
by nsim;
class treat;
var   treat  y;
monotone regression (y=  treat );
run;

ods listing;

data gentag;
   set  b(where=(treat =:'Xhigh' and dropout=100));
   by nsim;
   do _imputation_=1 to 100;

   output;
   end;
run;




proc print data=dataOutReg1 ;where nsim=10 and _imputation_=10;
run;


data datainReg2 ;
   set dataOutReg1 ;by nsim;
   if dropout<100 then y=y*(34-dropout)/34 + dropval;
run;

proc print data=datainReg2 ;where nsim=10 and _imputation_=10;
run;


data datainreg3;
   set datainReg2 gentag;
run;

proc sort data=datainreg3; by nsim _imputation_;
run;



proc sql;
   create table nodata as
       select nsim, treat, sum(dropout=0) as nodata, n(dropout) as n, calculated
nodata/calculated n as rate from datainreg3
      group by nsim, treat;
   select  treat, avg(rate) from  nodata
```

```
      group by  treat;
   quit;


ods listing exclude all;
proc mixed data = datainreg3;
by nsim _imputation_;
class treat ;
model y =    treat ;
lsmestimate treat 'treatdif' 1-1/CL;
ods output LSMestimates = estMI ;
run;


proc mianalyze data = estMI;
by nsim label;
modeleffects Estimate ;
stderr stderr;
ods output ParameterEstimates = MIana;
run;

ods listing;

proc sql;
   select sum(probt<0.05) as n_signif, n(probt) as n , avg(estimate) as mean_estimate from
miana;
   quit;
```

OUTPUT

dropout rate:

[redacted] The SAS System [redacted]

[redacted] 1

| treat | |
|---|---|
| Xhigh | 0.151229 |
| Xlow | 0.150564 |

proportion of  subjects without post-randomization data

[redacted] The SAS System [redacted]

[redacted] 6

| treat | |
|---|---|
| Xhigh | 0.151229 |
| Xlow | 0.150564 |

simulated power (100*n_signif/n)  and mean estimate



| n_signif | n | mean_ estimate |
|---|---|---|
| 7870 | 10000 | -2.11555 |